PharmaSUG 2013 - Paper SP02

Probability Based Criteria in Early Phase Drug Development

Howard Liang, inVentiv Health Clinical, Eden Prairie, Minnesota

ABSTRACT

Probability based criteria is in the study design where many research scientists prefer to understand the power and probability of taking a new drug forward across the whole range of possible true treatment effects, rather than focusing on one particular value to power the study. Examples will be used in this paper to show how to compute probability using the SAS®/STAT procedure PROC MIXED. Particular emphasis is given to the application on efficacy analysis, including the comparison to classical hypothesis testing. The application on safety analysis is also discussed.

INTRODUCTION

Probability based criteria in early phase drug development is based on the idea that the new drugs might do something but they might not be as good as the comparators on the market. It will help us to make correct decision before we start a study rather than making up the decision at the end of the study. The first example in this paper will explain how to set up the criteria on efficacy. The first part is to show separation to placebo, as in a standard power calculation. The second part is a harder hurdle designed to give confidence we are at least as good as standard of care. The second example in this paper will discuss the probability based on safety.

DATA SIMULATION

Suppose we have a simple clinical trial data with a sample of 50 subjects randomly assigned to placebo and a new drug with a 1:1 randomization ratio. The outcome is efficacy measured at four time points. The first measurement is a pre-dose baseline assessment which is followed by three post-dose repeated measurements. The following code produces the simulated efficacy data:

```
/** introduce variability **/
data sample;
  do subjid=101 to 110, 201 to 210, 301 to 310, 401 to 410, 501 to 510;
    if uniform(12345)>=0.5 then treatment='pl';
      else treatment='dx';
    do month=3, 2, 0, 1;
      result=10+0.5*(normal(6789));
      if month>0 and treatment='dx' then result=result-1.075;
      site=round(subjid,100.);
      output;
    end;
  end;
run;

/** derive change and baseline variables **/
proc sql;
  create table indata as
    select a.site, a.subjid, a.treatment, a.month,
           a.result-b.result as change, b.result as baseline
      from sample(where=(month>0)) as a, sample(where=(month=0)) as b
      where a.site=b.site and a.subjid=b.subjid
      order by site, subjid, month;
quit;
```

When we apply a simple mixed model repeated measurement (MMRM) analysis on the simulated efficacy data, the SAS®/STAT procedure PROC MIXED provides options to output least-squares means and difference of least-squares means with the ODS OUTPUT statement.

The following code creates the data sets LSM1 (least-squares means) and DIFF1 (difference of least-squares means):

```
ods output lsmeans=lsm1 diffs=diff1;
proc mixed data=indata;
    class site subjid month treatment;
```

```
    model change=site treatment month treatment*month baseline baseline*month/ddfm=kr;
    repeated month/subject=subjid type=un;
    lsmeans treatment*month/cl pdiff alpha=0.1;
run;
quit;
```

The data produced by PROC MIXED code above is stored in the work data sets LSM1 and DIFF1, and they can be combined to produce the following typical hypothesis testing report:

	Placebo	New Drug	Difference to Placebo	
Month	LS Mean(90% CI)	LS Mean(90% CI)	LS Mean(90% CI)	p-value
1	-0.07(-0.24,0.09)	-1.01(-1.18,-0.84)	-0.94(-1.18,-0.70)	<0.001
2	-0.02(-0.16,0.11)	-1.04(-1.18,-0.90)	-1.01(-1.21,-0.82)	<0.001
3	0.08(-0.07,0.24)	-1.20(-1.36,-1.04)	-1.28(-1.50,-1.06)	<0.001

Table 1. Typical Hypothesis Testing Report

In the analysis plan, we assume that:

"The sample size of 50 has been chosen to give 80% power to detect a 1-point difference from placebo using a 1-sided test with a 0.1 significance level."

Although there is a significant treatment difference from placebo in output 1, the significance test is 2-sided and compares against 0, not 1.

COMPUTING PROBABILITY BASED CRITERIA FOR EFFICACY APPLICATIONS

The choice to set up the probability based criteria usually depends on our knowledge about the compound, our clinical plan and the competition. In some therapeutic areas, the efficacy should be better than standard of care, so we can choose the first criteria to be "new drug is better than placebo with at least 90% probability" and the second criteria to be "new drug is better than standard of care with at least 67% probability". In other therapeutic areas, the efficacy is equivalent with the standard of care, but the new drug has a better safety profile and less side effects. In the second case we can choose the first criteria to be "new drug is better than placebo with at least 90% probability" and the second criteria to be "new drug is better than standard of care with at least 33% probability".

In this example, if the efficacy is equivalent with the standard of care, and the standard of care is about 1 reduction from the placebo, then following 2 part criteria would have been used:

Part 1 Criteria: New drug has a reduction > 0 with at least 90% probability

Part 2 Criteria: New drug has a reduction > 1 with at least 33% probability

Let X be the difference between the new drug and placebo, the probability is calculated as:

$$P(X<x) = probt[(x-d)/se, df]$$

where *d* is the estimated difference, *se* is the standard error of the estimated difference and *df* is the degrees of freedom associated with this difference.

So the following assignments would set up the 2 part criteria:

Part 1 Criteria: P (new drug – placebo < 0) = probt ((0 – *d*) / *se*, *df*) or

P (X<0) = prob0 = probt(-tvalue, df);

Part 2 Criteria: P (new drug – placebo < -1) = probt ((-1 – *d*) / *se*, *df*) or

P (X<-1) = prob1= probt(t1,df); where t1 = (-1-estimate)/se;

The following is the completed data step to create data set DIFF2:

```
data diff2;
     set diff1(where=( month=_month and  _treatment='pl'));
     cdiff=compress(put(estimate,6.2)||"("||put(lower,6.2)||","||put(upper,6.2)||")");
     pvalue=probt;
     prob0=compress(put(probt(-tvalue, df), 6.3));
     t1=(-1-estimate)/stderr;
     prob1=compress(put(probt(t1,df), 6.3));
 run;
```

The data set DIFF2 combined with LSM1 would have the following new report:

					Probability treatment
	Placebo	New Drug	Difference to Placebo		reduction > x(unit)
Month	LS Mean(90% CI)	LS Mean(90% CI)	LS Mean(90% CI)	p-value	x=0	x=1
1	-0.07(-0.24,0.09)	-1.01(-1.18,-0.84)	-0.94(-1.18,-0.70)	<0.001	1.000	0.334
2	-0.02(-0.16,0.11)	-1.04(-1.18,-0.90)	-1.01(-1.21,-0.82)	<0.001	1.000	0.549
3	0.08(-0.07,0.24)	-1.20(-1.36,-1.04)	-1.28(-1.50,-1.06)	<0.001	1.000	0.980

Table 2. Report Including the Probability Based Criteria on Efficacy

By achieving criteria 1, we are at least 90% sure that the new drug has some improvement over placebo. And by achieving criteria 2, we are at least 33% sure the new drug are better than standard of care or similar to competitor in efficacy. This means we can go forward with more confidence that we are in the region of what's already on the market, but further consideration is required for a favorable safety profile.

COMPUTING PROBABILITY BASED CRITERIA FOR SAFETY APPLICATIONS

In setting up probability based criteria for safety application, it is a little different than looking for confidence for efficacy. For example, we are looking for confidence we do not have an increase in a safety outcome with this class of drugs.

Suppose a drug with this safety parameter increase >5 is not good. To move forward to next phase of drug development, we want to achieve 'New drug effect on this safety parameter is <5 compared to placebo with at least 60% probability', so we have

P (new drug – placebo < 5) = probt ((5 – *d*) / *se*, *df*) or

P (X<5) = prob5 = probt(t5,*df*); where t5 = (5-estimate)/*se*;

By using following simulation safety data steps and PROC MIXED:

```
data sample;
  do subjid=101 to 110, 201 to 210, 301 to 310, 401 to 410, 501 to 510;
    if uniform(12345)>=0.5 then treatment='pl';
      else treatment='dx';
    do day=0,1,2,3,4,5,6,7,14,21,28;
      result=100+7.5*normal(6789);
      site=round(subjid,100.);
      output;
    end;
  end;
run;

proc sql;
  create table indata as
    select a.site, a.subjid, a.treatment, a.day,
           a.result-b.result as change, b.result as baseline
      from sample(where=(day>0)) as a, sample(where=(day=0)) as b
      where a.site=b.site and a.subjid=b.subjid
      order by site, subjid, day;
quit;
```

```
ods output lsmeans=lsm1 diffs=diff1;
proc mixed data=indata;
       class site subjid day treatment;
       model change=site treatment day treatment*day baseline baseline*day/ddfm=kr;
       repeated day/subject=subjid type=un;
       lsmeans treatment*day/cl pdiff alpha=0.1;
run;
quit;

data diff2;
     set diff1(where=( day=_day and  _treatment='pl'));
     cdiff=compress(put(estimate,6.2)||"("||put(lower,6.2)||","||put(upper,6.2)||")");
     pvalue=probt;
     t5=(5-estimate)/stderr;
     prob5=compress(put(probt(t5,df), 6.3));
run;
```

The data set DIFF2 combined with LSM1 would have the following new report:

	Placebo	New Drug	Difference to Placebo		Probability that Difference to
Day	LS Mean(90% CI)	LS Mean(90% CI)	LS Mean(90% CI)	P-value	Placebo < 5(unit)
1	-0.01(-2.84,2.82)	-0.53(-3.36,2.30)	-0.52(-4.55,3.50)	0.829	0.987
2	-0.06(-2.19,2.06)	-0.64(-2.76,1.49)	-0.57(-3.60,2.45)	0.752	0.998
3	-1.22(-3.26,0.82)	1.00(-1.04,3.05)	2.22(-0.69,5.14)	0.207	0.942
4	-3.82(-6.12,-1.51)	-1.12(-3.42,1.19)	2.70(-0.58,5.98)	0.174	0.877
5	-3.27(-5.89,-0.65)	0.86(-1.75,3.48)	4.13(0.41,7.86)	0.069	0.651
6	0.62(-1.84,3.07)	-2.33(-4.78,0.13)	-2.94(-6.44,0.56)	0.165	1.000
7	-0.40(-2.91,2.10)	0.42(-2.09,2.92)	0.82(-2.75,4.39)	0.701	0.972
14	-1.57(-3.97,0.82)	-0.80(-3.19,1.60)	0.78(-2.64,4.19)	0.704	0.978
21	-0.92(-3.22,1.39)	-1.85(-4.15,0.46)	-0.93(-4.21,2.35)	0.637	0.998
28	0.30(-2.03,2.62)	-0.33(-2.66,1.99)	-0.63(-3.94,2.68)	0.751	0.997

Table 3. Report Including the Probability Based Criteria on Safety

CONCLUSION

Probability based criteria have become more useful in the study design for research scientists in early phase drug development, because they provide a straightforward statistical framework that not only helps the study team to communicate and understand the new drug, but also enables the management team to make the correct decision. I hope this paper is useful to statistical analysts and others who perform statistical analysis duties in clinical trials or clinical research.

ACKNOWLEDGMENTS

I greatly appreciate to the contributions of Ying Chen and Nancy Brucken who reviewed the paper in draft and made many suggestions for improvement.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Howard Liang
Enterprise: inVentiv Health Clinical
Address: 12125 Technology Drive
City, State ZIP: Eden Prairie, MN 55344
Work Phone: (952) 833-7462
E-mail: howard.liang@inventivhealth.com
Web: http://www.inVentivHealthclinical.com